Strategic Interactions and MLE. (Lecture 13)

$$b_1 = \gamma \sum_{i \neq 1,\, i \in j} w_{1i} b_i + z_1 \phi + \varepsilon_1$$

$$b = \underbrace{\gamma Wb}_{\text{weighted combination of others' test scores, or other locations' taxes}} + \underbrace{z\phi}_{\text{indiv. controls}} + \underbrace{v}_{\text{location fixed effects}} + \varepsilon \text{ — errors.}$$

b: my test scores or my location's taxes

Can we estimate γ?

We certainly can't by OLS $b = \gamma Wb + z\phi + v + \varepsilon$

circular causality.

An IV that affects b_{-i} but not b?

<u>MLE</u>:

$$b - \gamma Wb = z\phi + v + \varepsilon$$

$$(1 - \gamma W) b = z\phi + v + \varepsilon$$

$$b = (1-\gamma W)^{-1} z\phi + (1-\gamma W)^{-1} v + (1-\gamma W)^{-1} \varepsilon$$

spatial correlation $\underbrace{\varepsilon}_{\text{my error}} = \varphi \underbrace{M\varepsilon}_{\text{errors of others in my group.}} + \xi$

$$(\varepsilon - \psi M \varepsilon) = \xi$$
$$(1 - \psi M)\varepsilon = \xi$$
$$\varepsilon = (1 - \psi M)^{-1} \xi$$

where $\xi \sim N(0, \sigma^2 I)$

$$b = (1-\gamma W)^{-1} Z \phi + (1-\gamma W)^{-1} V + (1-\gamma W)^{-1}(1-\psi M)^{-1} \xi$$

How does one estimate this? Note how this differs from

$$y = \alpha + \beta_1 x_1 + \beta_2 x_2 + \eta \quad \text{additive, linear}$$

→ MLE

- choose values of γ and ϕ, given b, W, Z, V, M and assuming $\xi \sim N(0, \sigma^2 \cdot I)$ that makes observed results the most probable.

Log likelihood for linear regression model (using normal dist.):

$$L = \prod_{i=1}^{N} \left(\frac{1}{\sigma} \underbrace{\phi\left(\frac{y_i - x_i\beta}{\sigma}\right)}_{\text{normal density function}} \right)$$

$$\log(L) = \sum_{i=1}^{N} \log \left(\frac{1}{\sigma} \phi\left(\frac{y_i - x_i \beta}{\sigma}\right)\right)$$

This is going to be something fractional. Maximizing something negative is like minimizing something positive.

If choose β to maximize $\log(L)$ given y_i, x_i, then minimizing errors essentially, because getting $|y_i - x_i\beta| = |\varepsilon|$ to be as small as possible...

e.g. $$y_i = \beta_1 X_{1i} + \beta_2 X_{2i} + \beta_3 X_{3i}^{\beta_4} + \beta_5 + \varepsilon_i$$

No way to do this with OLS $\Rightarrow$ use MLE

$$\ln(L_i) = \ln \phi\{(y_i - \theta_{1i} - \theta_{2i} X_{3i}^{\theta_{3i}})/\theta_{4i}\} - \ln \theta_{4i}$$

$$\theta_{1i} = \beta_1 X_{1i} + \beta_2 X_{2i} + \beta_5$$

$$\theta_{2i} = \beta_3$$

$$\theta_{3i} = \beta_4$$

$$\theta_{4i} = \sigma$$

```
*Linear MLE*
program drop _all
program mynormal_lf
 args lnf mu
 qui replace `lnf' = log(normalden($ML_y1-`mu'))
end

clear
set obs 100
set seed 12345
gen x=invnormal(uniform())
gen y=2*x + invnormal(uniform())
ml model lf mynormal_lf (y=x)()
ml maximize
reg y x


*Probit*
program drop _all
program myprobit_lf
 version 11
 args lnf xb
 qui replace `lnf' = ln(normal( `xb')) if $ML_y1 == 1
 qui replace `lnf' = ln(normal(-1*`xb')) if $ML_y1 == 0
end

clear
set obs 1000
set seed 12345
gen x = invnormal(uniform())
gen y = (0.5 + 0.5*x  > invnormal(uniform()))
ml model lf myprobit_lf (y = x)
ml maximize

probit y x

*Non-linear MLE*
program drop _all
program mynonlin_lf
version 11
args lnf theta1 theta2 theta3 sigma
qui replace `lnf'=ln(normalden($ML_y1,`theta1'+`theta2'*$X3^`theta3',`sigma'))
end

clear
set obs 100
set seed 12345
gen x1=invnormal(uniform())
gen x2=2*uniform()
gen x3=3*uniform()
gen y=2*x1+3*x2 +2*x3^2 + invnormal(uniform())
global X3 x3
ml model lf mynonlin_lf (y = x1 x2) /beta3 /beta4 /sigma
ml maximize
```